

Formação Desenvolvedor Moderno Módulo: Lógica de Programação

Capítulo: Estruturas Repetitivas https://devsuperior.com.br

1

## Estrutura repetitiva "enquanto"

https://devsuperior.com.br

Prof. Dr. Nelio Alves

## Estrutura repetitiva "enquanto"

É uma estrutura de controle que repete um bloco de comandos enquanto uma condição for verdadeira.

**Quando usar:** quando <u>não</u> se sabe previamente a quantidade de repetições que será realizada.

#### Problema exemplo:

Fazer um programa que lê números inteiros até que um zero seja lido. Ao final mostra a soma dos números lidos.

Digite o primeiro numero: 5

Digite outro numero: 2
Digite outro numero: 4
Digite outro numero: 0

SOMA = 11

3

## Sintaxe / regra

enquanto condição faca

comando 1

comando 2

fimenquanto

#### Regra:

V: executa e volta

F: pula fora

```
Algoritmo "teste_enquanto"

Var

x, soma : inteiro

Inicio

soma <- 0
escreva("Digite o primeiro numero: ")
leia(x)

enquanto x <> 0 faca
soma <- soma + x
escreva("Digite outro numero: ")
leia(x)
fimenquanto
escreval("SOMA = ", soma)

Fimalgoritmo
```


## Resumo da aula


- Estrutura repetitiva "enquanto"
- Recomendada quando não se sabe previamente a quantidade de repetições
- Regra:
  - V: executa e volta
  - F: pula fora


# Exercício propostos PARTE 1: testes de mesa com enquanto


https://devsuperior.com.br

Prof. Dr. Nelio Alves









## Estrutura repetitiva "para"

https://devsuperior.com.br

Prof. Dr. Nelio Alves

## Estrutura repetitiva "para"

É uma estrutura de controle que repete um bloco de comandos para um certo intervalo de valores.

Quando usar: quando se sabe previamente a quantidade de repetições, ou o intervalo de valores.

#### Por exemplo:

Fazer um programa que lê um valor inteiro N e depois N números inteiros. Ao final, mostra a soma dos N números lidos

```
Quantos números serão digitados? 3
Digite um numero: 5
Digite um numero: 2
Digite um numero: 4
SOMA = 11
```

13

## Sintaxe / regra

```
para variavel de valor_inicial ate valor_final [passo N] faca
  comando 1
  comando 2
fimpara
```

Regra primeira vez: A variavel é iniciada com o valor\_inicial

#### Regra de repetição:

Se o valor da variavel não exceder o valor\_final: executa e volta! Senão: pula fora!

**Regra toda vez que voltar:** incrementa a **variavel** de 1, ou do valor opcional em **passo** 

```
Algoritmo "teste_para"

Var

N, i, x, soma : inteiro

Inicio

escreva("Quantos numeros serao digitados? ")
leia(N)

soma <- 0
para i de 1 ate N faca
escreva("Digite um numero: ")
leia(x)
soma <- soma + x
fimpara
escreval("SOMA = ", soma)

Fimalgoritmo
```

## **Importante**

Perceba que a estrutura "para" é ótima para se fazer uma repetição baseada em uma CONTAGEM:

para i de 1 ate 5 faca
 escreval("Valor de i: ", i)
fimpara

Resultado na tela:

```
Valor de i: 1
Valor de i: 2
Valor de i: 3
Valor de i: 4
Valor de i: 5
```

## Contagem regressiva

```
para i de 5 ate 1 passo -1 faca
  escreval("Valor de i: ", i)
fimpara
```

#### Resultado na tela:

```
Valor de i: 5
Valor de i: 4
Valor de i: 3
Valor de i: 2
Valor de i: 1
```

17

### Resumo da aula

- Estrutura repetitiva "para"
- Usar quando se sabe previamente a quantidade de repetições
- Ótimo para fazer contagens (progressiva ou regressiva)
- Regra:

```
para variavel de valor_inicial ate valor_final [passo N] faca
comando 1
comando 2

fimpara

Regra primeira vez: A variavel é iniciada com o valor_inicial

Regra de repetição:
Se o valor da variavel não exceder o valor_final: executa e volta!
```


Senão: pula fora!


Regra toda vez que voltar: incrementa a variavel de 1, ou do valor opcional em passo


# Exercício propostos PARTE 3: testes de mesa com for


https://devsuperior.com.br

Prof. Dr. Nelio Alves









# Estrutura repetitiva "repita-até"

https://devsuperior.com.br

Prof. Dr. Nelio Alves

## Estrutura repetitiva "repita-até"

Menos utilizada, mas em alguns casos se encaixa melhor ao problema.

O bloco de comandos executa pelo menos uma vez, pois a condição é verificada no final.

25

## Sintaxe / regra

#### repita

comando 1 comando 2 ate condição

#### Regra:

V: pula fora F: volta

#### Problema exemplo:

Fazer um programa para ler uma temperatura em Celsius e mostrar o equivalente em Fahrenheit. Perguntar se o usuário deseja repetir (s/n). Caso o usuário digite "s", repetir o programa.

Fórmula: 
$$F = \frac{9C}{5} + 32$$

#### Exemplo:

```
Digite a temperatura em Celsius: 30.0
Equivalente em Fahrenheit: 86.0
Deseja repetir (s/n)? s
Digite a temperatura em Celsius: 21.0
Equivalente em Fahrenheit: 69.8
Deseja repetir (s/n)? s
Digite a temperatura em Celsius: -10.5
Equivalente em Fahrenheit: 13.1
Deseja repetir (s/n)? n
```

27

### Resumo da aula

- Estrutura repetitiva "repita-até"
- O bloco de comandos executa pelo menos uma vez, pois a condição é verificada no final.
- Regra:
  - V: pula fora
  - F: volta

repita

comando 1 comando 2 ate condição

```
Algoritmo "exemplo_repita_ate"

Var

C, F : real
    resp : caractere

Inicio

repita
    escreva("Digite a temperatura em Celsius: ")
    leia(C)
    F <- 9.0 * C / 5.0 + 32.0
    escreval("Equivalente em Fahrenheit: ", F:6:1)
    escreva("Deseja repetir (s/n)? ")
    leia(resp)
    ate resp <> "s"

Fimalgoritmo
```

## Solução dos exercícios:

github.com/acenelio/curso-algoritmos